

# Sistemas de ecuaciones

## 1 *Cómo introducir y resolver un sistema de ecuaciones lineales con MAXIMA:*

Se utiliza el desplegable

Ecuaciones>Resolver sistema lineal

Se introduce el sistema, ecuación por ecuación y se escriben las variables separadas por comas. Al indicar "aceptar" (o ejecutar la sentencia o evaluar la celda) se obtiene la solución. También se puede resolver con la instrucción

```
linsolve([lista de ecuaciones],[lista de incógnitas]).
```

EJEMPLO 1. Resolver el sistema  $x-y+z=1$ ,  $2x-y+z=0$ ,  $x+2y-z=0$ .

Usando el desplegable indicado, indicamos el número de ecuaciones: 3, e introducimos en cada hueco las ecuaciones  $x-y+z=1$ ,  $2x-y+z=0$ ,  $x+2y-z=0$  y las variables (o incógnitas):  $x,y,z$ . Al pulsar "aceptar" aparece:

```
(%i1) linsolve([x-y+z=1, 2*x-y+z=0, x+2*y-z=0], [x,y,z]);
(%o1) [x = - 1 , y = 3 , z = 5 ]
```

También se podría haber utilizado directamente la función:

```
linsolve ([expr_1, ..., expr_m], [x_1, ..., x_n])
```

## 2 *¿Cómo expresa Maxima que un sistema es incompatible.*

EJEMPLO 2. Resolver el sistema  $x+y=3$ ,  $2x+2y=7$ .

```
(%i2) linsolve([x+y=3, 2*x+2*y=7], [x,y]);
(%o2) []
```

La respuesta [] indica que Maxima no ha encontrado soluciones. Por lo tanto, el sistema es incompatible (no tiene solución).

## 3 *¿Cómo se expresa la solución dada para un sistema compatible indeterminado? ¿Cómo introduce Maxima los parámetros en la solución?*

EJEMPLO 3. Resolver el sistema  $x-z=2$ ,  $x-y+z=1$ ,  $3x-y-z=5$ .

```
(%i3) linsolve([x-z=2, x-y+z=1, 3*x-y-z=5], [x,y,z]);
solve: dependent equations eliminated: (3)
(%o3) [x = %r1 + 2, y = 2 %r1 + 1, z = %r1]
```

La solución dada nos indica que la tercera ecuación es eliminada porque es combinación de las anteriores.  
Los parámetros los ha introducido numerados automáticamente, con el símbolo % delante.  
En este caso, aparece sólo un parámetro y se denota por %r1, que coincide con z.

□ **4 ¿Cómo se resuelve un sistema con parámetros?**  
**OBSERVACIÓN: MAXIMA no diferencia incógnitas de de parámetros, ni distingue los posibles valores de los parámetros para los que no hay solución.**

✍ EJEMPLO 4. Resolver el sistema de parámetro a:  $ax-3y=5$ ,  $x+y=1$ .

✍ SOLUCIÓN.

```
(%i4) linsolve([a*x-3*y=5, x+y=1], [x, y]);
(%o4) [x = 8/(a+3), y = (a-5)/(a+3)]
```

✍ MAXIMA no distingue los distintos valores de un parámetro.  
Esto quiere decir, que debemos mirar la solución para interpretarla.

✍ ¿Cómo hay que interpretar la solución obtenida en el ejemplo anterior?  
Es claro que cuando el denominador es cero, los valores de x, y no existen.  
Como MAXIMA no distingue esa peculiaridad del valor  $a=-3$ , hay que entender que el sistema dado es compatible determinado cuando el valor de "a" es distinto de -3.  
Pero ojo, puede haber algún otro valor para el que el sistema no sea compatible determinado y que no aparezca en los denominadores de la solución porque se haya simplificado.  
Para el caso de valor  $a=-3$  debemos resolver el problema sustituyendo a por -3:

```
(%i5) linsolve([(-3)*x-3*y=5, x+y=1], [x, y]);
(%o5) []
```

✍ Como sabemos, la respuesta [] indica que no hay solución. Por lo tanto, el sistema es incompatible si  $a=-3$ .

✍ En vez de hacer la sustitución manual de a por -3, podemos dar un valor a un parámetro así:

```
(%i6) a:-3$
linsolve([a*x-3*y=5, x+y=1], [x, y]);
(%o7) []
```

✍ Observemos lo que ocurre en la siguiente ecuación con un parámetro a:

```
(%i8) kill(a);
      linsolve((a-2)*x=a^2-2*a,x);
(%o8) done
(%o9) [x = a]
```

Nótese que para  $a=2$  es compatible indeterminado y eso no lo detecta Maxima.

Un problema similar sucede si pedimos a Maxima que halle el rango de una matriz con un parámetro.

NOTA: El manejo de las matrices con Maxima se explica en otro documento.

```
(%i10) A:matrix([a+1,a],[1,2]);
      rank(A);
(%o10)  $\begin{bmatrix} a+1 & a \\ 1 & 2 \end{bmatrix}$ 
(%o11) 2
```

Obsérvese que el rango de A es 1 si  $a=-2$ .

La consecuencia que debemos sacar es que debe ser el usuario el que vigile estas posibles vicisitudes.

## 5 **MAXIMA puede resolver ecuaciones o sistemas generales, no necesariamente lineales:**

Para ello se selecciona el desplegable

Ecuaciones > Resolver,  
o bien, se usa la instrucción

```
solve([lista de ecuaciones],[lista de incógnitas]).
```

Ejemplos:

```
(%i12) solve([x^2-2*x=-4], [x]);
(%o12) [x = 1 -  $\sqrt{3}$  %i, x =  $\sqrt{3}$  %i + 1]
```

Esto quiere decir que la solución es compleja, el número imaginario  $i$  se respresenta por %i.

Cuando es una sola ecuación, los corchetes no son obligatorios. Además, si los segundos miembros de las ecuaciones son 0 no hace falta escribir '=0'. Ej: resolver  $4x+3=0$ , se indicaría:

```
(%i13) solve(4*x+3,x);
(%o13) [x = - $\frac{3}{4}$ ]
```

```
(%i14) solve([x^2-y^2=11,x-y=1],[x,y]);
(%o14) [[x = 6, y = 5]]
```

□ **6 Obsérvese que:**

**\*Si la solución de una ecuación no es real, MAXIMA la da como número complejo.**

**\*Es posible obtener sólo las soluciones reales. Véase Ecuaciones>Raíces reales de un polinomio.**

□ **7 Eliminación Gaussiana.**

Sea Amp la matriz ampliada asociada a un sistema de ecuaciones lineales.

La instrucción "echelon(Amp)" devuelve la forma escalonada de la matriz Amp obtenida por eliminación gaussiana.

Recordemos que la forma escalonada de una matriz se calcula mediante operaciones elementales con sus filas.

Las características de la forma escalonada son:

- el primer elemento no nulo de cada fila en la matriz escalonada es 1
- todos los elementos de la columna por debajo del primer 1 de cada fila son ceros.

La función "triangularize(Amp)" también lleva a cabo la eliminación gaussiana pero no normaliza el primer elemento no nulo de cada fila.

✓ (%i15) Amp:matrix([1,2,1,1],[1,0,-1,2],[2,4,0,3]);

(%o15) 
$$\begin{bmatrix} 1 & 2 & 1 & 1 \\ 1 & 0 & -1 & 2 \\ 2 & 4 & 0 & 3 \end{bmatrix}$$

✓ (%i16) triangularize(Amp);

(%o16) 
$$\begin{bmatrix} 1 & 0 & -1 & 2 \\ 0 & 2 & 2 & -1 \\ 0 & 0 & -4 & 2 \end{bmatrix}$$

✓ (%i17) echelon(Amp);

(%o17) 
$$\begin{bmatrix} 1 & 0 & -1 & 2 \\ 0 & 1 & 1 & -\frac{1}{2} \\ 0 & 0 & 1 & -\frac{1}{2} \end{bmatrix}$$

✓ La matriz anterior indica que en la solución del sistema es  $z = -1/2$ . Para calcular el valor de las otras variables hay que sustituir hacia atrás.

$y - 1/2 = -1/2$ , por lo tanto  $y = 0$ .

$x + 1/2 = 2$ , por lo tanto  $x = 3/2$ .

Si reescribimos el sistema equivalente y resolvemos resulta:

✓ (%i18) linsolve([x-z=2, y+z=-1/2, z=-1/2], [x,y,z]);

(%o18) 
$$\left[ x = \frac{3}{2}, y = 0, z = -\frac{1}{2} \right]$$

Podemos averiguar las operaciones elementales que ha realizado Maxima con las filas con el siguiente truco: añadimos a cada fila un parámetro que nos indique la fila que es y repetimos los calculos anteriores, así:

(%i19) `Amp:matrix([1,2,1,1,F1],[1,0,-1,2,F2],[2,4,0,3,F3]);`

(%o19) 
$$\begin{bmatrix} 1 & 2 & 1 & 1 & F1 \\ 1 & 0 & -1 & 2 & F2 \\ 2 & 4 & 0 & 3 & F3 \end{bmatrix}$$

(%i20) `triangularize(Amp);`

(%o20) 
$$\begin{bmatrix} 1 & 0 & -1 & 2 & F2 \\ 0 & 2 & 2 & -1 & F1 - F2 \\ 0 & 0 & -4 & 2 & 2 F3 - 4 F1 \end{bmatrix}$$

(%i21) `echelon(Amp);`

(%o21) 
$$\begin{bmatrix} 1 & 0 & -1 & 2 & F2 \\ 0 & 1 & 1 & -\frac{1}{2} & -\frac{F2 - F1}{2} \\ 0 & 0 & 1 & -\frac{1}{2} & -\frac{F3 - 2 F1}{2} \end{bmatrix}$$

Ahora vemos claro cuáles han sido las operaciones con las filas. Esto nos puede servir para comprobar con Maxima cálculos que hagamos a mano.

**8 Las acciones "construir la matriz de coeficientes y la matriz ampliada" se puede hacer con las instrucciones "coefmatrix" y "augcoefmatrix", respectivamente.**

La función "coefmatrix ([eqn\_1, ..., eqn\_m], [x\_1, ..., x\_n])" devuelve la matriz de coeficientes para las variables x\_1, ..., x\_n del sistema de ecuaciones lineales eqn\_1,..., eqn\_m.

La función "augcoefmatrix ([eqn\_1, ..., eqn\_m], [x\_1, ..., x\_n])" devuelve la matriz ampliada de coeficientes del sistema de ecuaciones lineales eqn\_1, ..., eqn\_m de variables x\_1, ..., x\_n construida igualando a cero las ecuaciones, es decir, los términos independientes se cambian de signo.

Ambas funciones son útiles para calcular los rangos de las matrices y poder clasificar el sistema de ecuaciones utilizando el teorema de Rouché-Frobenius..

Repetimos los razonamientos anteriores utilizando la matriz ampliada. Escribimos las ecuaciones iniciales del ejemplo 20.

```
(%i22) Amp:augcoefmatrix([x+2*y+z=1,x-z=2,2*x+4*y=3], [x,y,z]);
```

$$(\%o22) \begin{bmatrix} 1 & 2 & 1 & -1 \\ 1 & 0 & -1 & -2 \\ 2 & 4 & 0 & -3 \end{bmatrix}$$

```
(%i23) triangularize(%);
```

$$(\%o23) \begin{bmatrix} 1 & 0 & -1 & -2 \\ 0 & 2 & 2 & 1 \\ 0 & 0 & -4 & -2 \end{bmatrix}$$

Si ahora aplicamos la sustitución hacia atrás hay que tener en cuenta que los términos independientes están cambiados de signo. Sin embargo, si estamos calculando rangos no hay ningún problema con este cambio de signo.  
De cara a clasificar el sistema, en la forma escalonada, es claro que es compatible determinado.

## Otros ejercicios propuestos en el Curso Cero.

### **1 Ejercicio 14. (Página 40)**

Hay que clasificar y resolver un sistema dado, de 3 ecuaciones lineales y 3 incógnitas.

Con MAXIMA es muy fácil resolverlo:

```
(%i24) linsolve([2*x-y-2*z=1, x+z=2, x-y-z=3], [x,y,z]);
```

$$(\%o24) [x = 0, y = -5, z = 2]$$

Para clasificar el anterior sistema es suficiente ver que la solución es única: Es compatible determinado.

A la misma conclusión se habría llegado comparando los rangos de las matrices de coeficientes y de la matriz ampliada:

```
(%i25) A:coefmatrix([2*x-y-2*z=1, x+z=2, x-y-z=3], [x,y,z]);
```

$$(\%o25) \begin{bmatrix} 2 & -1 & -2 \\ 1 & 0 & 1 \\ 1 & -1 & -1 \end{bmatrix}$$

```
(%i26) rank(A);
```

$$(\%o26) 3$$

✓ (%i27) B:augcoefmatrix([2\*x-y-2\*z=1, x+z=2, x-y-z=3], [x,y,z]);

✓ (%o27) 
$$\begin{bmatrix} 2 & -1 & -2 & -1 \\ 1 & 0 & 1 & -2 \\ 1 & -1 & -1 & -3 \end{bmatrix}$$

✓ (%i28) rank(B);

✓ (%o28) 3

✓ También se podía haber hecho globalmente con la instrucción:

✓ (%i29) is(rank(coefmatrix([2\*x-y-2\*z=1, x+z=2, x-y-z=3], [x,y,z]))=rank(augcoefmatrix([2\*x-y-2\*z=1, x+z=2, x-y-z=3], [x,y,z])));

✓ (%o29) true

✓ Ambos rangos son iguales, e iguales al número de incógnitas, por tanto el sistema es compatible determinado.

## □ **2 Ejercicio 16. (Página 40)**

✓ (%i30) linsolve([-x-z=0,-x-2\*y+2\*z=0,x-y+z=0],[x,y,z]);

✓ (%o30) [x = 0, y = 0, z = 0]

✓ El sistema es homogéneo y sólo tiene la solución trivial.

## □ **3 Ejercicio 18. (Página 40)**

✓ De la misma forma que en los ejercicios anteriores hay que resolver un sistema de 3 ecuaciones con 3 incógnitas.

✓ (%i31) linsolve([x-y-2\*z=1, 2\*x-2\*y+z=0, x+y-z=0], [x,y,z]);

✓ (%o31)  $\left[x = -\frac{1}{10}, y = -\frac{3}{10}, z = -\frac{2}{5}\right]$

✓ La solución es única, por tanto es compatible determinado.

## □ **4 Recuerde seguir completando su fichero de instrucciones con los operadores de MAXIMA introducidos:**

✓ linsolve ([expr\_1, ..., expr\_m], [x\_1, ..., x\_n]) :Resuelve sistema de ecuaciones lineales

coefmatrix ([eqn\_1, ..., eqn\_m], [x\_1, ..., x\_n]) :Obtiene la matriz de coeficientes

augcoefmatrix ([eqn\_1, ..., eqn\_m], [x\_1, ..., x\_n]) :Obtiene la matriz ampliada tras cambiar el signo a los términos independientes.