

4. FUNCIONES DE UNA VARIABLE. EJERCICIOS

EJERCICIO 4.1. Sea $f: \mathbb{R} \rightarrow \mathbb{R}$ la función definida como:

$f(x) = x^2 + 2x + 1$, si $x \leq -1$; $f(x) = -x^2 + 1$, si $-1 < x \leq 1$; $f(x) = -2x + 2$, si $x > 1$.

a) Representétese gráficamente la función.

b) Estúdiase la continuidad de f .

c) Estúdiase la derivabilidad de f .

SOLUCIÓN. a) Para empezar, se define la función f :

```
(%i1) f1(x):=x^2+2*x+1; f2(x):=-x^2+1; f3(x):=-2*x+2;
```

```
(%o1) f1(x):=x^2+2*x+1
```

```
(%o2) f2(x):=-x^2+1
```


```
(%o3) f3(x):=(-2)*x+2
```

```
(%i4) f(x):=if x<=-1 then f1(x) else (if x<=1 then f2(x) else f3(x));
```

```
(%o4) f(x):=if x<=-1 then f1(x) else if x<=1 then f2(x) else f3(x)
```

Su representación gráfica en el intervalo $[-3, 3]$ es:

```
(%i5) wxplot2d([f(x)], [x,-3,3])$
```



b) Observando la gráfica, es claro que f es continua en todo \mathbb{R} . Aún así, vamos a estudiar la continuidad siguiendo la definición. Las funciones que definen a trozos a f son continuas, ya que son polinómicas, por lo que sólo es necesario estudiar la continuidad de f en los puntos $x = -1$ y $x = 1$.

En $x = -1$ se tiene

```
(%i6) f(-1);
(%o6) 0
```

Ahora hay que estudiar el límite de f por la izquierda y por la derecha de $x=-1$, respectivamente. Con Maxima no se pueden calcular límites de funciones definidas a trozos, por lo que hay que indicar explícitamente la función correspondiente a cada trozo. Así pues,

```
(%i7) limit(f1(x),x,-1,minus); limit(f2(x),x,-1,plus);
(%o7) 0
(%o8) 0
```

Los límites laterales coinciden con el valor 0, luego el límite cuando f tiende hacia -1 es 0. Como el valor del límite coincide con $f(-1)$, se deduce que f es continua en $x=-1$.

Razonando de modo análogo, se concluye que f es continua también en $x=1$.

c) Las funciones que definen a f a trozos son derivables por ser polinómicas, por lo que sólo hay que estudiar la derivabilidad en los puntos $x=-1$ y $x=1$.

Para $x=-1$, hay que comprobar si existe el límite cuando h tiende hacia 0 de $(f(-1+h)-f(-1))/h$. Puesto que f está definida de diferente forma a la izquierda y a la derecha de $x=-1$, se estudian los límites laterales:

```
(%i9) limit((f1(-1+h)-f(-1))/h,h,0,minus);
      limit((f2(-1+h)-f(-1))/h,h,0,plus);
(%o9) 0
(%o10) 2
```

Dado que los límites laterales no coinciden, se deduce que no existe $f'(-1)$.

Para el punto $x=1$:

```
(%i11) limit((f2(1+h)-f(1))/h,h,0,minus); limit((f3(1+h)-f(1))/h,h,0,plus);
(%o11) -2
(%o12) -2
```

Como los límites laterales coinciden, existe el límite cuando h tiende hacia 0 de $(f(1+h)-f(1))/h$ y vale -2 , es decir, f es derivable en $x=1$ y $f'(1)=-2$. Por tanto, se concluye que f es derivable en $\mathbb{R} \setminus \{-1\}$.

EJERCICIO 4.2. Determinéense los valores de a y b para que la función $f: \mathbb{R} \rightarrow \mathbb{R}$ definida como:

$$f(x) = ax^3 - 3x + 1, \text{ si } x \leq 2; \quad f(x) = 2x + b, \text{ si } x > 2;$$

sea derivable en \mathbb{R} .

SOLUCIÓN. Se define la función f :

```
(%i13) f1(x):=a*x^3-3*x+1; f2(x):=2*x+b;
      f(x):=if x<=2 then f1(x) else f2(x);
```

```
(%o13) f1(x):= a x^3 - 3 x + 1
```

```
(%o14) f2(x):= 2 x + b
```

```
(%o15) f(x):= if x <= 2 then f1(x) else f2(x)
```

Las funciones f_1 y f_2 son derivables en \mathbb{R} por ser polinómicas, luego la función f es derivable en \mathbb{R} salvo quizá en el punto $x=2$.

Por otra parte, para que f sea derivable en $x=2$, tiene que ser en particular continua, es decir, el límite de f cuando x tiende hacia 2 tiene que coincidir con $f(2)$.

Se calcula el valor de f y de los límites laterales en $x=2$:

```
(%i16) f(2); limit(f1(x),x,2,minus); limit(f2(x),x,2,plus);
```

```
(%o16) 8 a - 5
```

```
(%o17) 8 a - 5
```

```
(%o18) b + 4
```

Para que exista el límite cuando x tiende hacia 2 de $f(x)$, los límites laterales calculados tienen que coincidir con $f(2)=8a-5$. Por tanto, se debe cumplir que $8a-5=b+4$.

Se resuelve la ecuación $8a-5=b+4$ en función de las variables a y b . Se utiliza la instrucción `globalsolve:true` para asignar a las variables a y b la solución obtenida.

```
(%i19) globalsolve:true$
      solve([%o17-%o18],[a,b]);
```

```
(%o20) [[a:  $\frac{\%r1 + 9}{8}$ , b: %r1]]
```

Puesto que se trata de una ecuación con dos incógnitas, la solución aparece expresada en forma paramétrica, en este caso, en función del parámetro $\%r1$.

Además, para que f sea derivable en $x=2$, deben coincidir los límites laterales cuando h tiende hacia 0 de $(f(2+h)-f(2))/h$.

Se calculan dichos límites:

```
(%i21) limit((f1(2+h)-f(2))/h,h,0,minus);
      limit((f2(2+h)-f(2))/h,h,0,plus);
```

```
(%o21)  $\frac{3 \%r1 + 21}{2}$ 
```

```
(%o22) 2
```

Se igualan los límites y se resuelve la ecuación en función de la variable $\%r1$:

(%i23) solve(%o21-%o22);

(%o23) $\left[\%r1 = -\frac{17}{3} \right]$

Se calculan a y b en función del valor obtenido de %r1:


(%i24) a:ev(a,%r1=-17/3); b:ev(b,%r1=-17/3);

(%o24) $\frac{5}{12}$

(%o25) $-\frac{17}{3}$

Para los valores obtenidos, se representa la gráfica de f en el intervalo $[-3, 7]$:

(%i26) wxplot2d(f(x), [x,-3,7])\$



EJERCICIO 4.3. Determinése la recta tangente a la gráfica de la función $f(x)=x^4+3x^2-5$ en el punto $x=1$ y represéntese gráficamente la función y la tangente.

SOLUCIÓN. Se define la función f:

(%i27) f(x):=x^4+3*x^2-5;

(%o27) $f(x) := x^4 + 3x^2 - 5$

La ecuación de la recta tangente a f en $x=1$ viene dada por:
 $tgte(x)=f(1)+f'(1)(x-1)$:

(%i28) define(Df(x),diff(f(x),x));
define(tgte(x),f(1)+Df(1)*(x-1));


(%o28) $Df(x) := 4x^3 + 6x$

(%o29) $tgte(x) := 10(x-1) - 1$

En la siguiente figura se puede ver la representación gráfica de f y de la recta tangente a f en $x=1$.

(%i30) wxplot2d([f(x), tgte(x)], [x,-1,3])\$

(%t30)



EJERCICIO 4.4. Sea la función $f: \mathbb{R} \rightarrow \mathbb{R}$ definida como

$$f(x) = 5x^2 / (3x - 2).$$

Analícese y representese gráficamente la función.

SOLUCIÓN. Primero se define la función con Maxima:

(%i31) f(x):=5*x^2/(3*x-2);

(%o31) $f(x) := \frac{5x^2}{3x-2}$

Puesto que se trata de un cociente de polinomios, el dominio viene dado por el conjunto de los puntos x tales que el denominador no se anula, es decir, tales que x es distinto de $2/3$.

Estudiemos las asíntotas. En $x=2/3$ se tiene que

(%i32) limit(f(x), x, 2/3, minus); limit(f(x), x, 2/3, plus);

(%o32) $-\infty$

(%o33) ∞

Por tanto, hay una asíntota vertical en $x=2/3$. Al acercarse a esta asíntota por la izquierda la función tiende hacia $-\infty$ y al acercarse por la derecha tiende a $+\infty$.

Estudiemos la existencia de asíntotas oblicuas $y=mx+n$. En caso de que haya una asíntota oblicua cuando x tiende a $+\infty$, el valor de su pendiente m viene dado por

```
(%i34) limit(f(x)/x, x, inf);
(%o34)  $\frac{5}{3}$ 
```

y el valor de n:

```
(%i35) limit(f(x)-%*x, x, inf);
(%o35)  $\frac{10}{9}$ 
```

Puesto que el limite anterior es finito, existe una asíntota oblicua, cuya ecuación es $y = (5/3)x + 10/9$. Llamémosla $a(x)$:

```
(%i36) define(a(x),5/3*x+10/9);
(%o36)  $a(x) := \frac{5x}{3} + \frac{10}{9}$ 
```

Análogamente, la asíntota oblicua cuando $x \rightarrow -\infty$:

```
(%i37) limit(f(x)/x, x, minf);
(%o37)  $\frac{5}{3}$ 
```

```
(%i38) limit(f(x)-%*x, x, minf);
(%o38)  $\frac{10}{9}$ 
```

Por lo que se obtiene la misma asíntota oblicua cuando $x \rightarrow -\infty$.

Dado que f tiene una asíntota oblicua, no tiene asíntotas horizontales, pues la existencia de una excluye la existencia de las otras.

Estudiemos los intervalos de crecimiento y decrecimiento y los extremos relativos. Denotemos la derivada de f por Df :

```
(%i39) define(Df(x),diff(f(x),x,1));
(%o39)  $Df(x) := \frac{10x}{3x-2} - \frac{15x^2}{(3x-2)^2}$ 
```

Simplificamos:

```
(%i40) ratsimp(%);
(%o40)  $Df(x) := \frac{15x^2 - 20x}{9x^2 - 12x + 4}$ 
```

El dominio de Df viene dado por el conjunto de puntos donde no se anula el denominador. El denominador se anula en

```
(%i41) solve(9*x^2-12*x+4,x);
(%o41) [x = 2/3]
```

Por tanto, el dominio de Df es $(-\infty, 2/3)$ unión $(2/3, +\infty)$

Las raíces de Df son:

```
(%i42) solve(Df(x),x);
(%o42) [x = 4/3, x = 0]
```

Luego los puntos $x=4/3$ y $x=0$ son posibles extremos relativos.

Para estudiar el signo de la derivada en los intervalos que resultan: $(-\infty, 0)$, $(0, 2/3)$, $(2/3, 4/3)$ y $(4/3, +\infty)$, basta con comprobar el signo en un punto de cada uno de ellos:

```
(%i43) Df(-1); Df(1/3); Df(1); Df(2);
(%o43) 7/5
(%o44) -5
(%o45) -5
(%o46) 5/4
```

Así pues, Df es positiva en $(-\infty, 0)$ y $(4/3, +\infty)$ y negativa en $(0, 2/3)$ y $(2/3, 4/3)$, por tanto, f es creciente en $(-\infty, 0)$ unión $(4/3, +\infty)$ y decreciente en $(0, 2/3)$ unión $(2/3, 4/3)$. Luego en $x=0$ hay un máximo relativo y en $x=4/3$ hay un mínimo relativo. El valor del máximo relativo es

```
(%i47) f(0);
(%o47) 0
```

Y el valor del mínimo relativo es

```
(%i48) f(4/3);
(%o48) 40/9
```

Para determinar los intervalos de concavidad y convexidad se analiza la derivada segunda, que se denota con Df2:

```
(%i49) define(Df2(x),diff(f(x),x,2));
(%o49) Df2(x) := 10/(3*x-2) - 60*x/(3*x-2)^2 + 90*x^2/(3*x-2)^3
```

cuya expresión simplificada es:

(%i50) ratsimp(%);

(%o50) $Df2(x) := \frac{40}{27x^3 - 54x^2 + 36x - 8}$

Puesto que el numerador es $40 > 0$, para estudiar el signo de $Df2$ hay que analizar sólo el signo del denominador. Éste se anula en:

(%i51) solve(27*x^3-54*x^2+36*x-8, x);

(%o51) $[x = \frac{2}{3}]$

por lo que basta con evaluar $Df2$ en un punto cualquiera a la izquierda y a la derecha de $x = 2/3$:

(%i52) Df2(0); Df2(1);

(%o52) -5


(%o53) 40

Por tanto, $Df2$ es negativa en $(-\infty, 2/3)$ y es positiva en $(2/3, +\infty)$, luego f es cóncava en $(-\infty, 2/3)$ y convexa en $(2/3, +\infty)$.

El análisis realizado permite esbozar la gráfica de f . A continuación se representan f y las asíntotas con Maxima. Dado que la asíntota vertical no puede ser expresada por medio de una función, hay que utilizar coordenadas paramétricas para representarla. Utilizando la sentencia `['parametric, 2/3, t, [t, -20, 20]` se representa el conjunto de puntos de la forma $[2/3, t]$, con t variando entre -20 y 20, que pertenecen a la recta $x = 2/3$.

(%i54) wxplot2d([f(x), a(x), ['parametric, 2/3, t, [t, -20, 20],
[nticks, 300]], [x, -4, 4], [y, f(-4), f(4)])\$

plot2d: some values were clipped.



EJERCICIO 4.5. Obténgase el polinomio de Taylor de orden 7 ($p_7(x)$) de la función

$$f(x) = \ln(x)$$

en el punto $x=1$ y determínese una estimación del error cometido al aproximar f por p_7 en el punto $x=2$.

SOLUCIÓN. Definimos f :

```
(%i55) define(f(x), log(x));
```

```
(%o55) f(x) := log(x)
```

El polinomio de Taylor de orden 7 de f en el punto $x=1$ es

```
(%i56) define(p7(x), taylor(f(x), x, 1, 7));
```

```
(%o56)/T/ p7(x) := x - 1 - (x-1)^2/2 + (x-1)^3/3 - (x-1)^4/4 + (x-1)^5/5 - (x-1)^6/6 + (x-1)^7/7 + ...
```

El error cometido al aproximar f por p_7 en $x=2$ viene dado por el resto de Lagrange de orden 7 en $x=2$ (R_7), que es igual a

```
(%i57) define(Df8(x), diff(f(x), x, 8)); R7: (Df8(c)/8!)*(2-1)^8;
```

```
(%o57) Df8(x) := -5040/x^8
```

```
(%o58) -1/(8*c^8)
```

donde c es un valor perteneciente al intervalo $(1, 2)$. Así pues, el error cometido está comprendido entre los siguientes valores:

```
(%i59) float(-1/8); float(-1/(8*2^8));
```

```
(%o59) -0.125
```

```
(%o60) -4.8828125 10^-4
```

EJERCICIO 4.6 Determínese el polinomio de Mac Laurin de orden 5 ($p_5(x)$) de la función

$$f(x) = \sin(x)$$

y hállese el error cometido al aproximar f por p_5 en el punto $x=1$.

SOLUCIÓN: Se define f :

```
(%i61) define(f(x), sin(x));
```

```
(%o61) f(x) := sin(x)
```

El polinomio de Mac Laurin de orden 5 de f viene dado por

```
(%i62) define(p5(x), taylor(f(x), x, 0, 5));
```

```
(%o62)/T/ p5(x) := x - x^3/6 + x^5/120 + ...
```

✓ y la diferencia $f(1) - p_5(1)$ es

✓ (%i63) $f(1) - p_5(1);$
 [(%o63)/R/ $\frac{120 \sin(1) - 101}{120}$

✓ La expresión anterior vale

✓ (%i64) $\text{float}(\%);$
 [(%o64) $-1.9568185877017659 \cdot 10^{-4}$

✓ Por otra parte, la diferencia $f(1) - p_5(1)$ viene dada también por el resto de Lagrange de orden 5 en $x=1$ (R5). Se calcula la derivada sexta

✓ (%i65) $\text{define}(Df6(x), \text{diff}(f(x), x, 6));$
 [(%o65) $Df6(x) := -\sin(x)$

✓ Y el resto de Lagrange de orden 5 en $x=1$ es:

✓ (%i66) $R5: (Df6(c)/6!)*(1)^6;$
 [(%o66) $-\frac{\sin(c)}{720}$

✓ donde c es un valor comprendido entre 0 y 1. Se puede obtener el valor de c resolviendo

✓ (%i67) $\text{float}(\text{solve}(\%o66 - \%o64, c));$
 rat: replaced 1.9568185877017659E-4 by 159219/813662549 = 1.9568185877017673E-4
 solve: using arc-trig functions to get a solution.
 Some solutions will be lost.
 [(%o67) $[c = 0.14136127206865]$

✓ que efectivamente está comprendido entre 0 y 1.

✓ EJERCICIO 4.7. Se considera la ecuación $x + \sin(x) = 3$.
 a) Pruébese que tiene solución en el intervalo $[1, 3]$.
 b) Estúdiase si $g(x) = 3 - \sin(x)$ cumple el teorema del punto fijo en ese intervalo.
 c) En caso afirmativo, obténganse los cincuenta primeros valores de aproximación de la solución de la ecuación inicial por el método del punto fijo a partir del valor inicial $x_0 = 2$.

✓ SOLUCIÓN. a) Se define la función f del siguiente modo:

✓ (%i68) $\text{define}(f(x), x + \sin(x) - 3);$
 [(%o68) $f(x) := \sin(x) + x - 3$


La función $f(x)=x+\sin(x)-3$ verifica el Teorema de Bolzano, ya que f es continua en \mathbb{R} por ser suma de funciones continuas y

```
(%i69) float(f(1)*f(3));
(%o69) -0.1634916239615
```

es menor que 0. Así pues, la ecuación $f(x)=0$ tiene al menos una raíz en el intervalo $[1, 3]$.

Representando gráficamente f en el intervalo $[1, 3]$

```
(%i70) wxplot2d(f(x), [x,1,3]);
```



```
(%o70)
```

es claro que $f(x)=0$ tiene una única raíz en $[1, 3]$, que se corresponde con la coordenada x donde la gráfica de f corta al eje x .

b) Veamos primero que $g([1, 3])$ está contenido en $[1, 3]$. Para ello, hay que calcular el mínimo y el máximo absolutos de g en el intervalo $[1, 3]$. Como g es continua en ese intervalo, que es cerrado y acotado, g alcanza el máximo y el mínimo absolutos o bien en puntos críticos, o bien en los extremos del intervalo $[1, 3]$. Se define g y su derivada Dg :

```
(%i71) define(g(x), 3-sin(x)); define(Dg(x), diff(g(x),x));
(%o71) g(x):= 3 - sin(x)
(%o72) Dg(x):= - cos(x)
```

Los puntos críticos de g son:

```
(%i73) solve(Dg(x),x);
```

solve: using arc-trig functions to get a solution.
Some solutions will be lost.

```
(%o73) [ $x = \frac{\pi}{2}$ ]
```

Luego el único punto crítico es $x = \pi/2$, que pertenece a $[1, 3]$.
Se tiene que

```
(%i74) float(g(%pi/2)); float(g(1)); float(g(3));
```

```
(%o74) 2.0
```

```
(%o75) 2.158529015192103
```

```
(%o76) 2.858879991940133
```


por tanto, el mínimo absoluto de g se alcanza en $\pi/2$ y el máximo absoluto en 3 y como $g(\pi/2)$ y $g(3)$ pertenecen a $[1, 3]$ se cumple que $g([1, 3])$ está contenido en $[1, 3]$.

Además, es evidente que $Dg(x) \leq k$, para todo x en $(1, 3)$ y para algún k en $[0, 1)$, con lo que se satisfacen las hipótesis del teorema del punto fijo y g tiene un único punto fijo en $[1, 3]$.

Representando gráficamente la función g (en azul) y la bisectriz del primer cuadrante $y=x$ (en rojo), la solución (llamémosla c) de $f(x)=0$ es el valor de la coordenada x del punto de corte de g con la bisectriz, siendo las coordenadas de este punto $[c,c]$.

```
(%i77) wxplot2d([g(x),x], [x,-1,5],[y,-1,5],[legend, false],[gnuplot_preamble,
"set size ratio 1; set zeroaxis;"]);
```

```
(%t77)
```



```
(%o77)
```

c) Calculemos los cincuenta primeros valores de aproximación de la solución de la ecuación inicial por el método del punto fijo.

Generaremos una sucesión recurrente cuyo valor inicial es 2:

```
(%i78) c[1]:2; c[n]:=g(c[n-1]);
(%o78) 2
(%o79)  $c_n := g(c_{n-1})$ 
```

Y los cincuenta primeros valores de aproximación son:

```
(%i80) float(makelist(c[n],n,1,50));
(%o80) [2.0, 2.090702573174318, 2.132134239824455, 2.15345632800787,
2.164998051940865, 2.171404325029546, 2.175007839243745,
2.177049696277345, 2.178211417258738, 2.178873911586063,
2.179252208292936, 2.179468384405713, 2.179591970090704, 2.17966263999456,
2.179703056740294, 2.179726173277658, 2.179739395486308,
2.179746958528911, 2.179751284619006, 2.179753759180638,
2.179755174658586, 2.179755984330643, 2.179756447474479,
2.179756712399527, 2.179756863940608, 2.179756950624394,
2.179757000208836, 2.179757028571893, 2.179757044795996, 2.17975705407643,
2.179757059384981, 2.179757062421553, 2.179757064158519, 2.17975706515209,
2.179757065720428, 2.179757066045526, 2.179757066231487, 2.17975706633786,
2.179757066398707, 2.179757066433512, 2.179757066453421,
2.179757066464809, 2.179757066471324, 2.17975706647505, 2.179757066477182,
2.179757066478401, 2.179757066479098, 2.179757066479497,
2.179757066479725, 2.179757066479856]
```

En el valor de aproximación número 50, la función f vale

```
(%i81) float(f(c[50]));
(%o81) -7.4718009557273035 10-14
```

que es prácticamente 0, por lo que $c[50]$ es una muy buena aproximación a la raíz de $f(x)=0$ en el intervalo $[1, 3]$.


Ampliando la gráfica anterior, se puede ver cómo los puntos $[c[i], g(c[i])]$ se aproximan al punto $[c,c]$. Se dibujan sólo los 15 primeros puntos. Para representar puntos se emplea la sentencia `[discrete, puntosg]` y se utiliza la instrucción `[style,lines,lines,points]` para indicar que las funciones $g(x)$ e $y=x$ se representan de forma continua y los puntos de forma discreta, sin unirlos mediante líneas rectas.

```
(%i82) puntosg: float(makelist([c[n],g(c[n])],n,1,15));
```

```
(%o82) [[2.0,2.090702573174318],[2.090702573174318,2.132134239824455],[
2.132134239824455,2.15345632800787],[2.15345632800787,2.164998051940865
],[2.164998051940865,2.171404325029546],[2.171404325029546,
2.175007839243745],[2.175007839243745,2.177049696277345],[
2.177049696277345,2.178211417258738],[2.178211417258738,
2.178873911586063],[2.178873911586063,2.179252208292936],[
2.179252208292936,2.179468384405713],[2.179468384405713,
2.179591970090704],[2.179591970090704,2.17966263999456],[
2.17966263999456,2.179703056740294],[2.179703056740294,
2.179726173277658]]
```

```
(%i83) wxplot2d([g(x),x,[discrete, puntosg]], [x,1.95,2.25],[y,1.90,2.25], [style, lines, lines, points],
[legend, false],[gnuplot_preamble, "set size ratio 1; set zeroaxis;"]);
```

```
(%t83)
```




```
(%o83)
```

En la siguiente gráfica se muestra cómo los puntos $[c[i], f(c[i])]$ se aproximan al punto $[c, 0]$. Se representan sólo los 15 primeros puntos.

```
(%i84) puntosf: float(makelist([c[n],f(c[n])],n,1,15));
```

```
(%o84) [[2.0,-0.090702573174318],[2.090702573174318,-0.041431666650137],
[2.132134239824455,-0.021322088183415],[2.15345632800787,-
0.011541723932994],[2.164998051940865,-0.006406273088682],[
2.171404325029546,-0.0036035142141985],[2.175007839243745,-
0.0020418570335999],[2.177049696277345,-0.0011617209813931],[
2.178211417258738,-6.6249432732534697 10-4],[2.178873911586063,-
3.7829670687272721 10-4],[2.179252208292936,-2.1617611277657822 10-4],[
2.179468384405713,-1.2358568499126754 10-4],[2.179591970090704,-
7.0669903856623861 10-5],[2.17966263999456,-4.0416745733407033 10-5],[
2.179703056740294,-2.3116537364087364 10-5]]
```

```
(%i85) wxplot2d([f(x), [discrete, puntosf]], [x,2,2.2],[y,-0.1,0.01],
[style, lines, points],[legend, false]);
```



```
(%o85)
```

EJERCICIO 4.8. Dada la ecuación $x^3 - 2x - 5 = 0$,

- justifíquese que tiene al menos una solución en el intervalo $[2, 3]$,
- calcúlense los sesenta primeros valores de aproximación por el método de la bisección,
- estímese el error cometido al aproximar la raíz de la ecuación por el valor de aproximación número 60.

SOLUCIÓN. a) Se define la función f como:

```
(%i86) define(f(x), x^3-2*x-5);
```

```
(%o86) f(x) := x^3 - 2 x - 5
```


f es continua por ser polinómica y además

```
(%i87) f(2)*f(3);
```

```
(%o87) -16
```

es menor que 0, por lo que se satisface el teorema de Bolzano, lo que garantiza la existencia de al menos una solución de la ecuación $f(x) = 0$ en el intervalo $[2, 3]$. De hecho, representando gráficamente f

```
(%i88) wxplot2d(f(x), [x, 2,3]);
```



se ve que la gráfica de f corta a la recta $y=0$ en dicho intervalo, teniendo la ecuación $f(x)=0$ una única solución en $[2, 3]$.

b) Sean A y B los extremos del intervalo $[2, 3]$:

```
(%i89) A:2; B:3;
```

```
(%o89) 2
```

```
(%o90) 3
```

Con el siguiente bucle se genera la sucesión con las 60 primeras aproximaciones a la raíz de $f(x)=0$. Con él se ha programado el método de la bisección (aunque la programación con Maxima no forma parte del curso, las sentencias `for` e `if` empleadas son similares a las de la mayoría de los lenguajes de programación, por lo que nos permitimos la introducción de este programa a modo de un pequeño ejemplo de la programación con Maxima).

Inicializamos las variables: $a=A$, $b=B$.

```
(%i91) a:A; b:B;
```

```
(%o91) 2
```

```
(%o92) 3
```



```
(%i93) for i:1 thru 60 step 1 do
(
  if (f(a)*f(b)<0)
  then
  (
    c[i]:(a+b)/2,
    if (f(a)*f(c[i])<0)
    then b:c[i]
    else a:c[i]
  )
  else
  return(false)
);
```

```
(%o93) done
```

```
(%i94) fpprec:20; bfloat(makelist(c[i],i,1,60));
```

```
(%o94) 20
```

```
(%o95) [2.5b0, 2.25b0, 2.125b0, 2.0625b0, 2.09375b0, 2.109375b0, 2.1015625b0,
2.09765625b0, 2.095703125b0, 2.0947265625b0, 2.09423828125b0,
2.094482421875b0, 2.0946044921875b0, 2.09454345703125b0,
2.094573974609375b0, 2.0945587158203125b0, 2.09455108642578125b0,
2.094554901123046875b0, 2.0945529937744140625b0, 2.0945520401000976563b0,
2.0945515632629394531b0, 2.0945513248443603516b0,
2.0945514440536499023b0, 2.0945515036582946777b0, 2.09455147385597229b0,
2.0945514887571334839b0, 2.094551481306552887b0, 2.0945514850318431854b0,
2.0945514831691980362b0, 2.0945514822378754616b0,
2.0945514817722141743b0, 2.0945514815393835306b0,
2.0945514816557988524b0, 2.0945514815975911915b0,
2.0945514815684873611b0, 2.0945514815539354459b0,
2.0945514815466594882b0, 2.0945514815430215094b0, 2.09455148154120252b0,
2.0945514815421120147b0, 2.0945514815425667621b0,
2.0945514815423393884b0, 2.0945514815422257016b0, 2.094551481542282545b0,
2.0945514815423109667b0, 2.0945514815423251775b0, 2.094551481542332283b0,
2.0945514815423287303b0, 2.0945514815423269539b0,
2.0945514815423260657b0, 2.0945514815423265098b0,
2.0945514815423267319b0, 2.0945514815423266208b0,
2.0945514815423265653b0, 2.0945514815423265931b0,
2.0945514815423265792b0, 2.0945514815423265861b0,
2.0945514815423265896b0, 2.0945514815423265913b0,
2.0945514815423265922b0]
```

Evaluando f en la aproximación número 60, se comprueba que $f(c[60])$ es muy próximo a 0, es decir, $c[60]$ es un valor muy próximo a la raíz de $f(x)=0$.

```
(%i96) bfloat(f(c[60]));
```

```
(%o96) 8.118999313044612979b-18
```

c) El error cometido al aproximar la raíz c por $c[60]$ es menor que

```
(%i97) bfloat((B-A)/2^60);  
(%o97) 8.6736173798840354721b-19
```

Por otra parte, el comando `find_root` utiliza el método de la bisección y con él se obtiene una solución aproximada de $f(x)=0$:

```
(%i98) find_root(f(x),x,2,3); bfloat(%);  
(%o98) 2.094551481542327  
(%o99) 2.0945514815423265098b0
```

Como se puede comprobar, la solución que facilita `find_root` se corresponde con `c[51]` y el valor de f en esa solución es

```
(%i100) bfloat(f(%));  
(%o100) -9.115700815714999905b-16
```